


TEC LCD Cheat Sheet

Pin	Value	Function
1	VSS	GND
2	VDD	5V
3	V0	Contrast
4	RegSel	H=Data, L=Inst
5	R/W	H=Read, L=Write
6	Enable	H=Enable, L=Disable
7	DB0	Data Lines D0-D7
8	DB1	
9	DB2	
10	DB3	
11	DB4	
12	DB5	
13	DB6	
14	DB7	
15	BLA	Backlight 5V
16	BLK	Backlight GND

Pinout for 16x2 and 20x4 are identical



Instruction Port = 0x04

Data Port = 0x84

Instruction Register

One Byte commands to configure the LCD Screen

OUT (04), A

When LCD turns on or resets, screen defaults with 0x01, 0x06, 0x08 and 0x30

Hex	Description	Hex	Description	Hex	Description
0x01	Clear Screen, Cursor reset	0x0F	Display On, Cursor On and Blink	0x40	Set CGRAM Address Pos 1
0x02	Return Cursor to top left	0x10	Move Cursor one to the left		(Address from 40-7F)
0x04	Decrement Cursor on write	0x14	Move Cursor one to the right		
0x06	Increment Cursor on write	0x18	Shift Display to the left	0x80	Set Row 1, Col 1 DDRAM Address
0x05	Display to Shift Right	0x1C	Shift Display to the right	0xC0	Set Row 2, Col 1 DDRAM Address
0x07	Display to Shift Left	0x30	8-Bit, 1 Line, 5x8 dots	0x94	Set Row 3, Col 1 DDRAM Address (20x4)
0x08	Display Off, Cursor Off	0x34	8-Bit, 1 Line, 5x11 dots	0xD4	Set Row 4, Col 1 DDRAM Address (20x4)
0x0C	Display On, Cursor Off	0x38	8-Bit, 2 Line, 5x8 dots		(Address from 80-A7, C0-E7)
0x0E	Display On, Cursor On	0x3C	8-Bit, 2 Line, 5x11 dots		See Screen Layout Below

OUT (84), A
to write a
character to
DDRAM

Only write
while
LCD is not
busy

Screen Layout

DDRAM Address Counter with Bit 7 set

Pos	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
Row 1	80	81	82	83	84	85	86	89	88	89	8A	8B	8C	8D	8E	8F	90	91	92	93
Row 2	C0	C1	C2	C3	C4	C5	C6	C7	C8	C9	CA	CB	CC	CD	CE	CF	D0	D1	D2	D3
Row 3	94	95	96	97	98	99	9A	9B	9C	9D	9E	9F	A0	A1	A2	A3	A4	A5	A6	A7
Row 4	D4	D5	D6	D7	D8	D9	DA	DB	DC	DD	DE	DF	E0	E1	E2	E3	E4	E5	E6	E7

20x4

IN A, (04)
read busy flag (bit 7)

BF	AC	AC	AC	AC	AC	AC	AC	AC
----	----	----	----	----	----	----	----	----

Pos	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	Off Screen
Row 1	80	81	82	83	84	85	86	89	88	89	8A	8B	8C	8D	8E	8F	90-A7
Row 2	C0	C1	C2	C3	C4	C5	C6	C7	C8	C9	CA	CB	CC	CD	CE	CF	D0-E7

16x2

Writing a character to the screen will increase/decrease the Address Counter automatically

To move the cursor to Row 2, Column 10 do LD A, 0xC9 / OUT (04), A

For IN A, (04), If Bit 7 is set, then LCD is Busy. Other bits are the current Address Counter

Creating Custom Characters

CGRAM Address	Character In DDRAM
40	0
48	1
50	2
58	3
60	4
68	5
70	6
78	7

0	0	0	0	0	0	0	1	0x01
0	0	0	0	0	0	1	1	0x03
0	0	0	0	0	1	0	1	0x05
0	0	0	0	1	0	0	1	0x09
0	0	0	0	1	0	0	1	0x09
0	0	0	0	1	0	1	1	0x0B
0	0	0	1	1	0	1	1	0x1B
0	0	0	1	1	0	0	0	0x18

Up to 8 Custom
Characters can be
programmed

Use Bits 0 to 4 only

After each byte is written
CGRAM Address
increases

To display character use
0-7 in DDRAM


Use **OUT (04), A** to set the CGRAM address, where **A** is between 40h-7Fh
Then **OUT (84), A** to write one 5 pixel row

Character Table

Upper 4 Bits		Lower 4 Bits															
		0000	0001	0010	0011	0100	0101	0110	0111	1000	1001	1010	1011	1100	1101	1110	1111
xxxx0000	CG RAM (1)				0	a	P	`	P				-	9	E	o	p
xxxx0001	(2)			!	1	A	Q	a	9			■	ア	チ	4	ä	q
xxxx0010	(3)			"	2	B	R	b	r			「	イ	ウ	×	þ	ø
xxxx0011	(4)			#	3	C	S	c	s			」	ウ	テ	E	ε	∞
xxxx0100	(5)			\$	4	D	T	d	t			、	エ	ト	ト	μ	Ω
xxxx0101	(6)			%	5	E	U	e	u			・	オ	ナ	ユ	ε	Ü
xxxx0110	(7)			&	6	F	V	f	v			ヲ	カ	ニ	ヨ	ρ	Σ
xxxx0111	(8)			'	7	G	W	g	w			ア	チ	ヌ	ウ	g	π
xxxx1000	(1)			(8	H	X	h	x			イ	ウ	ホ	リ	フ	×
xxxx1001	(2))	9	I	Y	i	y			ウ	ケ	ル	ル	´	y
xxxx1010	(3)			*	:	J	Z	j	z			エ	コ	ハ	レ	j	チ
xxxx1011	(4)			+	;	K	C	k	c			オ	サ	ヒ	ロ	*	ア
xxxx1100	(5)			,	<	L	¥	l	l			カ	シ	フ	ワ	φ	ア
xxxx1101	(6)			-	=	M	I	m	}			ユ	ズ	ハ	コ	ト	÷
xxxx1110	(7)			■	>	N	^	n	÷			ヨ	セ	ホ	°	ア	
xxxx1111	(8)			/	?	O	_	o	+			ッ	リ	マ	°	ö	

Note: The user can specify any pattern for character-generator RAM.

Example using CGRAM and DDRAM



```
LCD_IR: EQU 04H ;Instruction Port
LCD_DR: EQU 84H ;Data Port

START:

; LCD set up
CALL CHECK_BUSY ;Check for LCD Busy
LD A,0CH ;Display On, Cursor Off
OUT (LCD_IR),A ;Send to Instruction Register (IR)
CALL CHECK_BUSY ;Check for LCD Busy
LD A,38H ;8-Bit, 2 Lines, 5x8 Characters
OUT (LCD_IR),A ;Send to Instruction Register (IR)

; Tell the LCD that next data will be to CGRAM
CALL CHECK_BUSY ;Check for LCD Busy
LD A,40H ;CGRAM entry
OUT (LCD_IR),A ;Send to Instruction Register (IR)

; Save multiple characters to CGRAM using lookup table and OUTI
LD BC,4000H | LCD_DR ;B=40 (8 letters), C=84 (port number)
LD HL,TABLE ;LCD custom character table

LOOP1:
CALL CHECK_BUSY ;Check for LCD Busy
OUTI ;Send (HL) to port C, Decrement B, Increment HL
JR NZ, LOOP1 ;OUTI sets zero if B=0

; Display first line text
CALL CHECK_BUSY ;Check for LCD Busy
LD A,82H ;Move Cursor to Row 1, Col 3
OUT (LCD_IR),A ;Send to Instruction Register (IR)

; Send multiple characters using lookup table and OUTI
LD BC,0A00H | LCD_DR ;B=0A (10 letters), C=84 (port number)
LD HL,TEXT ;LCD text table

LOOP2:
CALL CHECK_BUSY ;Check for LCD Busy
OUTI ;Send (HL) to port C, Decrement B, Increment HL
JR NZ, LOOP2 ;OUTI sets zero if B=0

; Display new characters
CALL CHECK_BUSY ;Check for LCD Busy
LD A,0C0H ;Move Cursor to Row 2, Col 1
OUT (LCD_IR),A ;Send to Instruction Register (IR)

; Send multiple characters using lookup table and OUTI
LD BC,0800H ;B=08 (letters), C=0 (CGRAM ref)

LOOP3:
CALL CHECK_BUSY ;Check for LCD Busy
LD A,C ;Send Custom Character
OUT (LCD_DR),A ;Send to Data Register (DR)
INC C ;Increase C to next CGRAM ref in DDRAM
CALL CHECK_BUSY ;Check for LCD Busy
LD A,20H ;Send Space Character
OUT (LCD_DR),A ;Send to Data Register (DR)
DJNZ LOOP3 ;Repeat for all 8 characters

HALT ;Pause the CPU

TEXT:
DB "HELLO TEC!"

TABLE:
DB 00H,0AH,1FH,1FH,0EH,04H,00H,00H ;Heart
DB 04H,0EH,0EH,0EH,1FH,00H,04H,00H ;Bell
DB 1FH,15H,1FH,1FH,0EH,0AH,1BH,00H ;Alien
DB 00H,01H,03H,16H,1CH,08H,00H,00H ;Tick
DB 01H,03H,0FH,0FH,0FH,03H,01H,00H ;Speaker
DB 01H,03H,05H,09H,09H,0BH,1BH,18H ;Note
DB 00H,0EH,15H,1BH,0EH,0EH,00H,00H ;Skull
DB 0EH,11H,11H,1FH,1BH,1BH,1FH,00H ;Lock

; Routine to check the Busy Flag (Bit 7)
CHECK_BUSY:
IN A,(LCD_IR) ;Read Busy Byte
RLCA ;Put Bit 7 in Carry Flag
JR C, CHECK_BUSY ;If Carry Set then LCD is busy
RET
```